This first scan creates the average line.  Scan the universe of stocks, make sure you exclude preferreds, bonds, etc. You only want common shares. Adjust the volume filter according to the way volume is stored. Composites are created for each price band to reflect differences in volatility vs. price.
------------------------------------------------------------------------

Buy=0;

Filter=MA(V,20)>500; //500 here is 50,000 shares
past=Ref(C,-3);

Change=log(C/past);

str=Filter*IIf(past<2,2,IIf(past<5,5,IIf(past<10,10,IIf(past<20,20,IIf(past<40,40,100)))));

//AddColumn(str,"str",1.0);
//AddColumn(Change,"change",1.5);
flag=3;

AddToComposite((str==2)*Change,"~2","C",flag);

AddToComposite((str==2),"~2","V",flag);

AddToComposite((str==5)*Change,"~5","C",flag);

AddToComposite((str==5),"~5","V",flag);

AddToComposite((str==10)*Change,"~10","C",flag);

AddToComposite((str==10),"~10","V",flag);

AddToComposite((str==20)*Change,"~20","C",flag);

AddToComposite((str==20),"~20","V",flag);

AddToComposite((str==40)*Change,"~40","C",flag);

AddToComposite((str==40),"~40","V",flag);

AddToComposite((str==100)*Change,"~100","C",flag);

AddToComposite((str==100),"~100","V",flag);

------------------------------------------------------------------------

This is the second scan.  It creates the average-low and average-high values.  Scan the same common shares as in above. Do not run this scan until the first is completed.
------------------------------------------------------------------------

Buy=0;

Filter=MA(V,20)>500; //500 here is 50,000 shares
past=Ref(C,-3);

Change=log(C/past);

A2=Foreign("~2","C",1)/Foreign("~2","V",1);

A5=Foreign("~5","C",1)/Foreign("~5","V",1);

A10=Foreign("~10","C",1)/Foreign("~10","V",1);

A20=Foreign("~20","C",1)/Foreign("~20","V",1);

A40=Foreign("~40","C",1)/Foreign("~40","V",1);

A100=Foreign("~100","C",1)/Foreign("~100","V",1);

str=IIf(past<2,2,IIf(past<5,5,IIf(past<10,10,IIf(past<20,20,IIf(past<40,40,100)))));

strx=IIf(past<2,A2,IIf(past<5,A5,IIf(past<10,A10,IIf(past<20,A20,IIf(past<40,A40,A100)))));

up=Filter*(Change>strx);

down=Filter*(Change<strx);

AddToComposite((str==2)*up*Change,"~2u","C",3);

AddToComposite((str==2)*down*Change,"~2d","C",3);

AddToComposite((str==2)*up,"~2u","V",3);

AddToComposite((str==2)*down,"~2d","V",3);

AddToComposite((str==5)*up*Change,"~5u","C",3);

AddToComposite((str==5)*down*Change,"~5d","C",3);

AddToComposite((str==5)*up,"~5u","V",3);

AddToComposite((str==5)*down,"~5d","V",3);

AddToComposite((str==10)*up*Change,"~10u","C",3);

AddToComposite((str==10)*down*Change,"~10d","C",3);

AddToComposite((str==10)*up,"~10u","V",3);

AddToComposite((str==10)*down,"~10d","V",3);

AddToComposite((str==20)*up*Change,"~20u","C",3);

AddToComposite((str==20)*down*Change,"~20d","C",3);

AddToComposite((str==20)*up,"~20u","V",3);

AddToComposite((str==20)*down,"~20d","V",3);

AddToComposite((str==40)*up*Change,"~40u","C",3);

AddToComposite((str==40)*down*Change,"~40d","C",3);

AddToComposite((str==40)*up,"~40u","V",3);

AddToComposite((str==40)*down,"~40d","V",3);

AddToComposite((str==100)*up*Change,"~100u","C",3);

AddToComposite((str==100)*down*Change,"~100d","C",3);

AddToComposite((str==100)*up,"~100u","V",3);

AddToComposite((str==100)*down,"~100d","V",3);

------------------------------------------------------------------------

Now that the averages have been created, we need to interpolate between the price bands.  The results are the variables labeled ‘mm’ (average), ‘mu’ (average-high), and ‘ml’ (average-low).
------------------------------------------------------------------------

M2=Foreign("~2","C",1)/Foreign("~2","V",1);

U2=Foreign("~2u","C",1)/Foreign("~2u","V",1);

L2=Foreign("~2d","C",1)/Foreign("~2d","V",1);

M5=Foreign("~5","C",1)/Foreign("~5","V",1);

U5=Foreign("~5u","C",1)/Foreign("~5u","V",1);

L5=Foreign("~5d","C",1)/Foreign("~5d","V",1);

M10=Foreign("~10","C",1)/Foreign("~10","V",1);

U10=Foreign("~10u","C",1)/Foreign("~10u","V",1);

L10=Foreign("~10d","C",1)/Foreign("~10d","V",1);

M20=Foreign("~20","C",1)/Foreign("~20","V",1);

U20=Foreign("~20u","C",1)/Foreign("~20u","V",1);

L20=Foreign("~20d","C",1)/Foreign("~20d","V",1);

M40=Foreign("~40","C",1)/Foreign("~40","V",1);

U40=Foreign("~40u","C",1)/Foreign("~40u","V",1);

L40=Foreign("~40d","C",1)/Foreign("~40d","V",1);

M100=Foreign("~100","C",1)/Foreign("~100","V",1);

U100=Foreign("~100u","C",1)/Foreign("~100u","V",1);

L100=Foreign("~100d","C",1)/Foreign("~100d","V",1);

xc=Ref(C,-3);

mp=IIf(xc<2,2,IIf(xc<5,5,IIf(xc<10,10,IIf(xc<20,20,IIf(xc<40,40,100)))));

//inx=Cum(log(Foreign(".SPX","C",1)/Ref(Foreign(".SPX","C",1),-1)));
sm=IIf(xc<2,(M5-M2)/2,IIf(xc<5,(M10-M2)/6,IIf(xc<10,(M20-M5)/10,IIf(xc<20,(M40-

M10)/20,IIf(xc<40,(2*M100-M40-M20)/40,0)))));

bm=IIf(xc<2,(M5+M2)/2,IIf(xc<5,(M10+M5)/2,IIf(xc<10,(M20+M10)/2,IIf(xc<20,(M40+M20)/2,M100))));

mm=(xc-mp)*sm+bm;

su=IIf(xc<2,(U5-U2)/2,IIf(xc<5,(U10-U2)/6,IIf(xc<10,(U20-U5)/10,IIf(xc<20,(U40-

U10)/20,IIf(xc<40,(2*U100-U40-U20)/40,0)))));

bu=IIf(xc<2,(U5+U2)/2,IIf(xc<5,(U10+U5)/2,IIf(xc<10,(U20+U10)/2,IIf(xc<20,(U40+U20)/2,U100))));

mu=(xc-mp)*su+bu;

sl=IIf(xc<2,(L5-L2)/2,IIf(xc<5,(L10-L2)/6,IIf(xc<10,(L20-L5)/10,IIf(xc<20,(L40-

L10)/20,IIf(xc<40,(2*L100-L40-L20)/40,0)))));

bl=IIf(xc<2,(L5+L2)/2,IIf(xc<5,(L10+L5)/2,IIf(xc<10,(L20+L10)/2,IIf(xc<20,(L40+L20)/2,L100))));

ml=(xc-mp)*sl+bl;

------------------------------------------------------------------------

The code above needs to be included with the following code bits.  You can add with the ‘#include’ preprocessor command.  Here is an indicator to draw bands around the closing price.
------------------------------------------------------------------------

x=log(C/xc);

xx=IIf(x>mm,(x-mm)/(mu-mm),(x-mm)/(mm-ml))/2;

Plot(C,"",colorBlack,styleBar);

Plot(EMA(Ref(C,-3)*exp(mm),3),"",colorRed,1);

Plot(EMA(Ref(C,-3)*exp(ml),3),"",colorBlue,1);

Plot(EMA(Ref(C,-3)*exp(mu),3),"",colorBlue,1);

------------------------------------------------------------------------

To plot an oscillator, add:
------------------------------------------------------------------------

x=log(C/xc);

xx=IIf(x>mm,(x-mm)/(mu-mm),(x-mm)/(mm-ml))/2;

xy=EMA(xx,5);

Plot(xy,"",colorRed,1);

Plot(EMA(xy,3),"",colorBlue,1);
------------------------------------------------------------------------

If you want to see how the distribution of relative ROCs looks, change the variable ‘xc’ in the interpolation code from: xc=Ref(C,-3); to: xc=Param("Price",5,.5,50,.1); and add the following:
------------------------------------------------------------------------

Plot(EMA(mm,5),"",colorRed,1);

Plot(EMA(mu,5),"",colorBlue,1);

Plot(EMA(ml,5),"",colorBlue,1);

Plot(0,"",colorBlack,1);

------------------------------------------------------------------------

Don’t forget to bring up the slider bar with Ctrl-R. You will notice as price increases, the relative rate-of-change bands narrow.
